
options nofmterr; 
ods html close; *STOPS WRITING TO THE CURRENT RESULTS VIEWER; 
 
ods html; *OPENS A NEW RESULTS VIEWER; 
 
 /**********************************************************   
  
The directory in the libname below is unique to the download. 
The user needs to use their own directory where the data are stored. 
  
************************************************************/ 
libname sasfile 'P:\Documents\Documents\CALARIE - Rochon\data from website2\sas files'; 
  
/**************************************  
Date: 8/29/17 
Programmer: Carl Pieper 
Sample analysis - differences by CR group & gender in VO2max  
We will need the following variables..... 
PERSTRSS (outcome), time (visit), gender, CR group 
  
Looking at the Proc Contents, these variables are located in the sas data sets vo2max, IVRSRAND,  & 
SUBJECT1,  
  
******************************************************/ 
  
  
  
/*************************************;   
  
**** STEP 1: this program imports the data from a SAS xpt file *** 
  
libname xptfile xport 'P:\Documents\Documents\CALARIE - Rochon\data from 
website\sas_database\analysis.xpt' access=readonly; 
proc copy inlib=xptfile outlib=sasfile; 
run; 
  
*************************************** */ 
  
/**************************************  
* STEP 2:  READ in the formats; 
  
libname library 'P:\Documents\Documents\CALARIE - Rochon\data from website\sas_database'; 
proc format cntlin=sasfile.formats library=library; 
run; 



  
  
****************************************/ 
title1; 
  
* formats for visit; 
proc format; 
value visit 5='0: baseline (visit 5)' 
            9='6: Month 6 (visit 9)' 
                        11='12: Month 12 (visit 11)' 
                        13='24: Month 24 (visit 13)'; 
  
 value $tx 
         'A'='A: CR' 
         'B'='B: AL'; 
run; 
  
  
  
  
************************  Part 1: Main effects model - using variables measured at 0, 6, 12, 24 
months.  Standard Model ******************; 
/*  
  
proc print data=sasfile.ivrsrand; run*/ 
; 
  
proc sort data=sasfile.ivrsrand; by deidnum; run; 
proc sort data=sasfile.vo2max; by deidnum visit; run; 
proc sort data=sasfile.subject1; by deidnum; run; 
  
  
* step 1: gather all the Person level data; 
  
data person; 
 merge sasfile.ivrsrand (keep=deidnum tx in=intx) 
       sasfile.subject1 (keep=deidnum female in=indemog); 
           by deidnum; 
 in_tx=intx; 
 in_demog=indemog; 
  
 if in_tx; 
 run; 
  



 /*    
 * This is a check of screened vs. randomized; 
                 proc freq data=person; 
                  tables in_tx*in_demog/list missing; 
                  run; 
                 */ 
 /*  
 look at 10 VO2max records 
  
 proc contents data=sasfile.vo2max; run; 
 proc print data=sasfile.vo2max (obs=10); run; 
 proc freq data=sasfile.vo2max;  
   tables visit; 
 run; 
  
 */ 
  
  
 * add the person-visit level data; 
 data person_time; 
   merge person (in=inperson) 
         sasfile.vo2max (in=invo2 keep=deidnum visit PVOMEAS1 PVOMEAS2); 
                 by deidnum; 
 in_vo2=invo2; 
 in_person=inperson; 
  
 if in_vo2 and in_person; 
format visit visit. tx $tx.; 
  
if visit=5 then month=0; 
if visit=11 then month=12; 
if visit=13 then month=24; 
  
vo2max=mean(pvomeas1,pvomeas2); 
                 run; 
  
title1 **** a look at the number of observations at each wave  for VO2 ****; 
proc freq data=person_time; 
                 tables visit*in_person*in_vo2 tx/list missing; 
                 run; 
  
  
  
  



  
/* Draw pictures :  VO2 By Gender, Wave, and CR group ***** */ 
  
proc sort data=person_time out=forplots; 
  by female month tx; 
  run; 
  
  
  
proc means noprint data=forplots; 
 by female month tx; 
 var vo2max; 
 output out=plotmeans mean=; 
 run; 
  
 title1 VO2 By Gender and Wave; 
 proc sgplot data=plotmeans; 
 by female; 
 series y=vo2max x=month /group=tx; 
 run; 
  
  
******* MODELS ***************; 
  
  
  
Title1 full model 3-way interaction of visit, CR group, & gender; 
proc mixed data=person_time; 
 class visit (ref="0: baseline (visit 5)")   tx (ref="B: AL"); 
 model vo2max=tx|visit|female / solution; 
 repeated visit / subject=deidnum type=un rcorr; 
 run; 
 ****** NOTICE TYPE=UN Error structure assumed ***; 
  
  
Title1 reduced model - no 3 way interactino vist, CR, and Gender; 
proc mixed  data=person_time; 
 class visit (ref="0: baseline (visit 5)")   tx (ref="B: AL"); 
 model vo2max=tx|visit tx|female female|visit / solution type=cs rcorr ; 
 repeated visit / subject=deidnum; 
 run; 
****** NOTICE TYPE=CS Error structure assumed ***; 
  
title1 reduced 2 - Treatment by Gender only: VO2; 



proc mixed data=person_time; 
 class visit (ref="0: baseline (visit 5)")   tx (ref="B: AL"); 
 model vo2max=visit tx|female / solution; 
 repeated visit / subject=deidnum type=cs rcorr ; 
 run; 
  
  
  
  
  
title1 ************* CHANGE SCORE ANALYSIS *******************************************; 
  
 ************************  Part 2:  Modeling CHANGE in Vo2: 0-6, 0-12, 0-24 ***************; 
 ************************  show an example of 'controlling for CR' 
*********************************************; 
  
  
  
  
/***********  Adding PERCENT CR ******************* 
  
*********** look at PCT CR estimates ***************; 
  
  
proc print data=sasfile.pctcr (obs=10); run; 
proc contents data=sasfile.pctcr; run; 
  
proc format library=library fmtlib; 
   select  visfmt; 
run; 
***********************************/ 
  
/*** Step 1:  look at the CHANGE  in vo2  */ 
 data vo2_0 (drop=visit)  
          vo2_12 (drop=visit) 
          vo2_24 (drop=visit); 
          set sasfile.vo2max; 
  
           vo2max=mean(PVOMEAS1, PVOMEAS2); 
  
          if visit=5 then output vo2_0; 
          if visit=11 then output vo2_12; 
          if visit=13 then output vo2_24; 
run; 



  
  
  
data PCT_CR (drop=interval); 
set sasfile.pctcr (keep=deidnum interval pctcr); 
  
if interval=2 then do; period='12-0'; output; end; 
if interval=4 then do; period='24-0';output; end; 
run; 
  
  
  
  
  
  
data person2 (keep=period delta_vo2max vo2max_baseline tx female deidnum); 
merge vo2_0 (rename=(vo2max=vo2max_baseline) in=in0) 
      vo2_12 (rename=(vo2max=vo2max_12) in=in12) 
      vo2_24 (rename=(vo2max=vo2max_24) in=in24) 
      sasfile.ivrsrand (keep=deidnum tx in=in_rand) 
      sasfile.subject1 (keep=deidnum female);  
          format tx $tx.; 
by deidnum;  
  
in_0=in0; 
in_12=in_12; 
in_24=in24; 
in=in_rand; 
  
if in_rand; 
  
          * 12 mo. change; 
          delta_vo2max=vo2max_12-vo2max_baseline; 
          period='12-0'; 
          output; 
          * 24 mo. change; 
          delta_vo2max=vo2max_24-vo2max_baseline; 
          period='24-0'; 
          output; 
          run; 
  
proc sort data=person2; by deidnum period; run; 
proc sort data=pct_cr; by deidnum period; run; 
  



  
  
* LAST STEP:  Merge on the PCT CR data; 
         data person3; 
         merge person2 (in=inperson) pct_cr (in=inpctcr); 
           by deidnum period; 
           in_person=inperson; in_pctcr=inpctcr; 
           run; 
  
  
  
title2 - dump of 1st 20 subjects; 
proc print data=person3 (obs=20); run; 
  
proc sort data=person3 out=meanslook; by tx period; 
proc means data=meanslook; 
by tx period;  
run; 
  
  
  
/* plot of the results */ 
proc sort data=person3 out=plotit; 
by female period tx; 
run; 
title2 plot of CHANGE in VO2; 
proc means noprint data=plotit; 
 by female period tx;  
  var delta_vo2max; 
  output out=plotmeans mean=; 
  run; 
proc sgplot data=plotmeans; 
   by female; 
   series y=delta_vo2max x=period/group=tx; 
   run; 
  
  
* analysis of change scores - controlling for baseline; 
* one fully interactive;  
title2 Fully interactive model - control for baseline VO2; 
proc mixed data=person3; 
 class period (ref="12-0")   tx (ref="B: AL"); 
 model delta_vo2max=tx|period|female vo2max_baseline / solution; 
 repeated period / subject=deidnum type=un rcorr; 



 run; 
  
 title2 test of 2-way interactions  - controlling for baseline; 
proc mixed data=person3; 
 class period (ref="12-0")   tx (ref="B: AL"); 
 model delta_vo2max=tx|period tx|female period|female vo2max_baseline / solution; 
 repeated period / subject=deidnum type=un rcorr; 
 run; 
  
    
 title2 main effects only - controlling for PCT CR; 
 title2 AND CONTROLLING FOR PCT CR OBSERVED; 
proc mixed data=person3; 
 class period (ref="12-0")   tx (ref="B: AL"); 
 model delta_vo2max=tx period female vo2max_baseline tx|pctcr/ solution; 
 repeated period / subject=deidnum type=un rcorr; 
 run; 


